
Abstract
Despite the simple, one-dimensional nature of strings, several computationally hard problems on
strings are known. Tackling hard problems beyond sizes of toy instances with straight-forward
solutions is infeasible. To solve these problems on datasets of even small sizes, effort has to be
put into the conception of algorithms leveraging profound characteristics of the input. Finding
these characteristics can be eased by rapidly creating and evaluating prototypes of new concepts
in how to tackle hard problems. Such a rapid-prototyping method for hard problems is answer set
programming (ASP). In this light, we study the application of ASP on five NP-hard optimization
problems in the field of strings. We provide MAX-SAT and ASP encodings, and empirically reason
about the merits and flaws when working with ASP solvers.

1 Introduction

Despite the fact that most string problems found in literature are solvable in polynomial
time or even close to linear time or beyond, there are several problems that are known to
be NP-hard. Among those, we focus on five problems that are well-perceived regarding
the number of publications studying these problems: Closest String (csp)1, Closest
Substring (css), Longest Common Subsequence (lcs), Minimum Common String
Partition (mcsp), and Shortest Common Superstring (scs). These problems have
been studied under various viewpoints. With respect to fixed-parameter tractability (FPT),
Bulteau et al. [9] gave a comprehensive survey on various NP-hard problems related to
strings; this survey comprises the problems studied in this paper. Also, Basavaraju et al. [2]
studied the kernelization of a majority of our problems. We address other related work in
the individual sections of each problem, but omit references to approximation algorithms
due to their amount, and because we put focus on the exact solution of the aforementioned
problems formulated as optimization problems.

1 We stick to the commonly used abbreviation csp in literature despite that cs would fit better with the
abbreviations of the other problems.

17:2 Encoding Hard String Problems with Answer Set Programming

A major problem in tackling these problems in practice is that naive solutions quickly
become impractical with respect to the time complexity. Tailored algorithms2 are hard
to implement, and thus a burden on the algorithm engineering side. Our contribution is
to advertise answer set programming (ASP) as a rapid-prototype programming tool for
solving NP-hard string problems on small instances. ASP is a declarative programming
language geared towards solving hard problems [40, 12]. ASP has been successfully applied
in robotics [3], or for computing the n-queens and the knight’s tour problem [18]. There is
also a competition on ASP solvers on various classic problems addressing mainly problems
on graphs [28]. See [19, 20] and the references therein for an overview of other use cases.

Although well-devised algorithms can outperform ASP-based approaches, the program-
ming effort for writing in an expressive, declarative programming language such as ASP is
considerably small. In this paper, we devise MAX-SAT encodings for the above addressed
problems, and subsequently translate these encodings into the ASP language. With respect
to tackling hard string problems via MAX-SAT encodings we are aware of the work of Bannai
et al. [1] who studied MAX-SAT encodings for repetitiveness measures that are also known
to be NP-hard.

2 Preliminaries

Common to all problems treated in this paper is the input of a set of m strings S =
{S1, . . . , Sm}. For simplicity, we assume that all strings have the same length n, and that
all characters are drawn from an alphabet Σ of size σ = |Σ|. Hence, |Sx| = n denotes the
length of each input string and Sx[i] ∈ Σ for all i ∈ [1..n] and x ∈ [1..m]. Except for mcsp,
the output is a string T that is object to an optimization argument with respect to the input
strings (and, additionally for css, with respect to an integer parameter specifying the length
of T ).

Encoding Annotations. Beginning with the next section, we state rules and constraints
with numbered equations, and add to each equation, in square brackets, the number of
generated clauses and the size of each such clause. For instance, the equation

[O(n), O(1)] ∀i ∈ [1..n] : pi =⇒ pi+1 (1)

defines n clauses, each of the form (¬pi ∨ pi+1), so its complexity is [O(n), O(1)].

Experiments. We implemented our MAX-SAT-formulations in the ASP language, and
used the solver clingo [26, 27]3 for evaluation. We compare the results with brute-force
approaches written in the python language on randomly generated data. Our filenames
are formatted like s03m04n005i1 to denote that the alphabet size is σ = 3, the number of
strings is m = 4, the length of each string is n = 5, and this file is the i = 1-st sample of
a batch of files with the same characteristics (σ, m and n). For mcsp, we have file formats
like 2s02n008i2.txt where the prefix 2 denotes that m = 2 is fixed. For the mcsp files, we
assume that the two strings given have the same Parikh vector. Our implementations and
datasets are available at https://github.com/koeppl/aspstring. For the evaluation, all
experiments ran single-threaded on a machine with Intel Core i3–9100 CPU and Debian 11.

2 Meaning that such algorithms usually are based on theoretical results that can be put hardly into
practice.

3 https://github.com/potassco/clingo

https://github.com/koeppl/aspstring
https://github.com/potassco/clingo

S1 = l n e e p l e s s n e l s

S2 = s l e e p s l s s n e s n

S3 = n l e l p l e s s n s s s

S4 = s n e e p l e l s n s s s

S5 = s l l e e l e s s n s s s

S1 = l n e e p l e s s n e l s

S2 = s l e e p s l s s n e s n

S3 = n l e l p l e s s n s s s

S4 = s n e e p l e l s n s s s

S5 = s l l e e l e s s n s s s

T = s l e e p l e s s n e s s

Figure 1 Example for csp (Sect. 3) with n = 13. The input set S = {S1, . . . , S5} is shown on
the left figure. The right figure shows that the solution T = sleeplessness has three mismatches
with each of the input strings in the Hamming distance. Mismatching characters are highlighted by
surrounding boxes.

3 Closest String Problem (CSP)

The Closest String Problem (csp)4 asks for a string T such that
maxx∈[1..m] distham(Sx, T ) is minimal, where the Hamming distance distham is given
by distham(Sx, T ) := |{i ∈ [1..n] : Sx[i] ̸= T [i]}|. An example is shown in Fig. 1. Here, and
in the following examples we stick to the alphabet Σ := {e, l, p, n, s} with size σ = 5.

Related Work. Frances and Litman [24] and Lanctôt et al. [39] proved that csp and its
generalization, the Closest Substring Problem (css), are NP-hard for any alphabet
with σ ≥ 2 in n and m. The parameterized complexities have been surveyed in [48, Section
5.1] and [57], with focus also on css. For the decision problem with a Hamming distance of d,
Gramm et al. [32] showed that csp can be solved in O(mn + dd) time or 22O(m log m) O(log n)
time. Regarding integer linear programming (ILP), Chimani et al. [13] gave ILP formulations,
also for css. There is a line of research on further practical ILP formulations [16, 43, 54].
Finally, Knop et al. [38] gave also an ILP formulation and an exact algorithm running in
mO(m2) O(log n) time.

With respect to different kinds of optimization approaches, Kelsey and Kotthoff [37]
studied csp as a constraint satisfaction problem, Huan et al. [35] provided an ant colony
optimization algorithm, and Vilca and de Freitas [55] gave a specialized algorithm for fixed
m = 3.

3.1 MAX-SAT encoding
We use the known fact that we have to select, for the i-th character of the output T , a
character appearing at the i-th position of one of the input strings.

▶ Lemma 1 ([37, Lemma 2]). For each i ∈ [1..n], T [i] = Sx[i] for an x ∈ [1..m].

Let us define Σi := {S1[i], . . . , Sm[i]} to be the set of characters appearing at text
position i of all input strings. Then σi := |Σi| ≤ min(m, σ), and σi can be much less than
m or σ if the number of distinct characters is small. We can express the alphabets per
position Σi by a Boolean matrix M [1..n][1..σ] with M [i][c] = 1 if c ∈ Σi.

4 Alternative names are, among others, Minimum Radius, Center String or Consensus String
problem.

17:4 Encoding Hard String Problems with Answer Set Programming

Further, we define the variables Ti,c = 1 to encode that T [i] = c, for i ∈ [1..n], c ∈
{S1[i], . . . , Sm[i]}. To state that T [i] = Sx[i], we want that, for a fixed position i ∈ [1..n],
only one Ti,c is set:

[O(n), O(min(m, σ))] ∀i ∈ [1..n] :
∑
c∈Σi

Ti,c = 1 (CSP1)

Next, we define the cost variables Ci,x for all i ∈ [1..n] and x ∈ [1..m] with Ci,x being set if
T [i] ̸= Sx[i]. Thus the Hamming distance between T and Sx is distham(T, Sx) =

∑
i∈[1..n] Ci,x.

Therefore:

[O(nmσ), O(1)] ∀i ∈ [1..n], c ∈ Σi, x ∈ [1..m] : Ti,c ∧ Sx[i] ̸= c =⇒ Ci,x (CSP2)

A statement for setting Ci,x to false is not needed as the optimizer will try to do so if it
does not violate (CSP2). This is achieved by the following objective:

[O(1), O(mn)] minimize max
x∈[1..m]

∑
i∈[1..n]

Ci,x (CSP3)

Complexities. We have O(nσ) selectable variables (Ti,c), O(nm) helper variables (Ci,x),
O(nmσ) clauses (CSP2). The largest clause contains O(mn) variables (CSP3).

Implementation. Our implementation in ASP is given in Listing 1. In all listings, the
percent sign % introduces a comment until the end of the line, which we use to refer to the
MAX-SAT equation that is represented by the respective line of code. Red curly arrows
symbolize line breaks. If not otherwise stated, in all code listings onwards, we assume that
the input is of the form s(X, I, C), denoting that SX[I] = C ∈ Σ. We use the helper variables
mat(X,I) to denote the existence of SX[I]. For encoding (CSP3) in ASP, we additionally
define the helper variables cost and mcost encoding

∑
i∈[1..n] Ci,X and maxx∈[1..m] cost(x),

respectively. The #show directives at the end define the variables the solver has to output.
The evaluation for our implementation is deferred until we have introduced the css problem,
which we conjointly evaluate in Sect. 4.2.

Listing 1 ASP for csp (Sect. 3).

mat(X,I) :- s(X,I,_).
1 {t(I,C) : s(_,I,C)} 1 :- mat(_,I). %(CSP1)
c(X,I) :- t(I,C), s(X,I,A), C != A. %(CSP2)
cost(X,C) :- C = #sum {1,I : c(X,I)}, mat(X,_). %(CSP3)
mcost(M) :- M = #max {C : cost(_,C)}.
#minimize {M : mcost(M)}.
#show t/2. #show mcost/1. #show cost/2.

4 Closest Substring (CSS)

For the css problem, we additionally require a parameter λ as input to specify the length of
the output string T . css asks for the string T with |T | = λ such that maxx∈[1..m] distλ(Sx, T )
is minimal, where distλ(Sx, T ) := mini∈[1..n−λ+1] distham(Sx[i..i + λ − 1], T ) is the number
of mismatches we need to be able to detect T via approximate pattern matching in Sx with
distλ(Sx, T ) mismatches. An example is shown in Fig. 2.

1 2 3 4 5 6 7 8 9 10 11 12 13

S1 = s l e s n l e s s p e s s

S2 = s n e l p e l l n e s s s

S3 = s s s s s s l p s p e s s

S4 = p s e l n e s e e l s e s

S5 = n e s s s l s n e l e s s

1 2 3 4

S1 = s l e s n l e s s p e s s

S2 = s n e l p e l l n e s s s

S3 = s s s s s s l p s p e s s

S4 = p s e l n e s e e l s e s

S5 = n e s s s l s n e l e s s

T = s n e s

Figure 2 Example for css (Sect. 4) with n = 13 and query length λ = 4. The input is shown on
the left figure. We can observe in the right figure that T = snes is the css having one mismatch
with each of the input strings in the Hamming distance by horizontally shifting the input strings.

Related Work. The decision problem for δ mismatches is also called δ-Mismatch prob-
lem. Gramm et al. [32, Theorem 2] solved the decision problem in O(mλ + (n −
λ)mδδ+1) time. Marx [46] showed that css can be solved in O(σδ(lg δ+2(nm)O(lg δ)) or
O((σδ)O(mδ)(nm)O(log log m)) time. A survey on further results can be found in [31]. With
respect to other optimization approaches, we are aware of a genetic algorithm [47].

4.1 MAX-SAT encoding
Following [32, Section 3.3], we reduce css to csp by selecting shifts dx ∈ [0..n − λ] of each
input string Sx such that the csp of {S1[1 + d1..λ + d1], . . . , Sm[1 + dm..λ + dm]} is a solution
of css if we take the minimum distance over all shifts dx.

In what follows, we represent the shifts by a matrix of selectable Boolean variables of
size O(m(n − λ)). We redefine the alphabet for the i-th character to be Σi := {S1[i +
d1], . . . , Sm[i + dm]}. We define the variables Ti,c and Ci,x as before. We copy (CSP1) as it
is since it only states from which string Sx we select the i-th character of T , except that we
have O(λ) instead of O(n) clauses since |T | = λ. The major difference is that for checking
equality, we must add the offsets and obtain the following modification of (CSP2):

[O(λnmσ), O(1)] ∀i ∈ [1..λ], c ∈ Σi, x ∈ [1..m] : Ti,c ∧ Sx[i + dx] ̸= c =⇒ Ci,x (CSS2)

The additional n-term in the complexity stems from the fact that the offsets dx are represented
as a two-dimensional binary array. The other equations as well as the objective are kept in
the same way.

Complexities. We have O(λσ + m(n − λ)) selectable variables (Ti,c and dx), O(λm) helper
variables (Ci,x), O(λmnσ) clauses. The largest clause has size O(λm). Our implementation
in ASP is given in Listing 2, where we expect an additional input of the form #const
lambda=λ. for the requested substring length λ.

Listing 2 ASP for css (Sect. 4).

mat(X,I) :- s(X,I,_).
1 {d(X,D) : D = 0..n-lambda} 1 :- mat(X,0).
sigma(I,C) :- s(X,J,C), d(X,D), J-D >= 0, I = J-D.
1 {t(I,C) : sigma(I,C)} 1 :- mat(_,I), I < lambda. %(CSP1)
c(X,I) :- t(I,C), s(X,J,A), d(X,D), I+D == J, I < lambda, A != C. %(CSS2)
cost(X,C) :- C = #sum {1,I : c(X,I)}, mat(X,_). %(CSP3)
mcost(M) :- M = #max {C : cost(_,C)}.
#minimize {M : mcost(M)}.
#show t/2. #show mcost/1. #show cost/2.

CPM 2023



17:6 Encoding Hard String Problems with Answer Set Programming

Table 1 Evaluation for the Closest String Problem (csp) for λ = 0 and Closest Substring
Problem (css) for λ > 0. The column dist shows the maximum Hamming distance of the reported
string to all input strings. The column rules is the number of created SAT rules, vars is the number
of variables, and choices is the number of choices or configurations the solver or brute-force algorithm
tries. Reported times are in seconds ([s]).

ASP brute-force
file λ dist rules vars choices time [s] choices time [s]
s05m09n009i0 0 6 1288 321 725 0.01 640 000 5.47
s05m09n009i0 7 4 1932 1122 1663 0.02 78 125 1.96
s05m09n009i0 8 5 1764 969 3666 0.05 390 625 7.29
s05m09n009i0 9 6 1427 330 676 0.01 1 953 125 21.59
s06m07n009i1 0 7 1078 268 1767 0.02 768 000 5.12
s06m07n009i1 7 4 1765 1069 3235 0.04 279 936 5.49
s06m07n009i1 8 5 1550 868 1314 0.02 1 679 616 24.45
s06m07n009i1 9 7 1194 275 2058 0.02 10 077 696 87.80
s06m08n009i0 0 6 1191 295 1074 0.01 750 000 5.67
s06m08n009i0 7 5 1907 1147 4266 0.05 279 936 6.23
s06m08n009i0 8 6 1698 954 4021 0.05 1 679 616 27.90
s06m08n009i0 9 6 1319 303 1273 0.01 10 077 696 100.23
s06m08n009i1 0 7 1248 299 2378 0.02 1 800 000 13.63
s06m08n009i1 7 5 1971 1203 4834 0.07 279 936 6.27
s06m08n009i1 8 6 1770 1012 5093 0.08 1 679 616 27.77
s06m08n009i1 9 7 1380 307 2163 0.02 10 077 696 99.98
s06m08n009i2 0 7 1248 299 2128 0.02 1 800 000 13.61
s06m08n009i2 7 5 1907 1147 5556 0.07 279 936 6.28
s06m08n009i2 8 6 1698 955 5552 0.08 1 679 616 27.91
s06m08n009i2 9 7 1380 307 2210 0.02 10 077 696 99.84
s06m09n009i0 0 7 1303 322 1837 0.02 800 000 6.81
s06m09n009i0 7 4 2142 1301 4331 0.05 279 936 7.02
s06m09n009i0 8 5 1920 1093 5334 0.08 1 679 616 31.38
s06m09n009i0 9 7 1443 331 1962 0.02 10 077 696 111.16
s06m09n009i1 0 7 1396 328 1849 0.02 2 700 000 22.97
s06m09n009i1 7 5 2177 1334 5341 0.07 279 936 7.04
s06m09n009i1 8 6 1920 1100 5693 0.10 1 679 616 31.20
s06m09n009i1 9 7 1542 337 1746 0.02 10 077 696 110.05
s06m09n009i2 0 6 1336 324 1874 0.02 1 080 000 9.07
s06m09n009i2 7 4 2177 1333 3706 0.05 279 936 6.92
s06m09n009i2 8 5 1946 1114 4565 0.06 1 679 616 30.52
s06m09n009i2 9 6 1478 333 1920 0.02 10 077 696 107.62

4.2 Evaluation of csp and css
Although there are efficient heuristics like choosing a majority string [8], we compared
our ASP encoding for csp to a basic brute-force algorithm that enumerates all possible
assignments for the characters of the closest substring. The number of possible configurations
for T is cS :=

∏n
i=1 σi ∈ O(min(σn), mn) dependent on the shape of the strings in S. A

brute-force algorithm trying each configuration spends O(cSnm) time on computing the
Hamming distances of the resulting string T with all strings of S.

This algorithm can be easily adopted for css. For that, we consider all possible offsets
of the input strings like in the ASP encoding. Hence, the number of configurations is the
number of configurations for the csp instance, multiplied by (n − λ)m for each possible
offset value. If λ is small, then it suffices to compute all configurations of T , which are σλ

many, and compute the Hamming distances in O(λm) time for each such configuration. We
implemented the former brute-force approach, whose time complexity grows exponentially
with all parameters σ, n, and m, for randomly generated strings. We can observe this case
in Table 1, where the ASP implementation outperforms the brute-force approach.

Table 2 Evaluation of the Closest String problem (scp) on datasets provided by Torres and
Hoshino [54]. The column distance is the maximal Hamming distance of the output to any of the
input strings.

file distance rules vars choices time [s]
rand-4-150-150-5-2 2 31 329 12 942 19 0.06
rand-4-50-50-5-2 2 10 529 4342 21 0.015
rand-4-100-100-5-2 2 20 929 8642 24 0.031
rand0-2-10-10-20-5 4 4286 842 43 0.011
rand0-2-10-10-20-4 4 4286 842 60 0.011
rand0-4-10-10-20-5 4 4887 1179 65 0.012
rand0-2-10-10-20-3 5 4474 848 72 0.012
rand-20-50-50-5-2 2 17 323 4549 78 0.018
rand-20-150-150-5-2 2 55 819 13 197 100 0.082
rand0-20-10-10-20-5 4 5573 1359 121 0.013
rand-20-100-100-5-2 2 37 213 8894 129 0.041
rand-4-150-150-5-1 5 31 329 12 942 189 0.117
rand-4-50-50-5-1 5 10 529 4342 199 0.021
rand0-2-10-10-20-2 6 4474 922 202 0.014
rand-4-100-100-5-1 5 20 929 8642 248 0.056
rand0-2-10-10-20-1 7 4474 922 265 0.015
rand-4-50-50-10-2 5 12 279 3082 494 0.035
rand-20-100-100-5-1 5 37 213 8894 501 0.068
rand-20-150-150-5-1 5 55 819 13 197 525 0.131
rand-20-50-50-5-1 5 18 595 4585 548 0.029
rand0-4-10-10-20-4 5 5008 1264 555 0.018
rand0-4-10-10-20-3 5 4869 1241 627 0.019
rand0-20-10-10-20-4 5 5800 1384 998 0.027
rand-20-50-50-10-2 5 26 397 3511 1057 0.053
rand0-20-10-10-20-3 6 6520 1477 2369 0.058
rand-20-50-50-15-2 7 40 426 5288 3512 0.235
rand-4-50-50-15-2 7 18 454 4622 4192 0.251
rand-4-50-50-10-1 8 12 279 3082 7320 0.343
rand0-4-10-10-20-2 8 5255 1334 18 095 0.373
rand-20-50-50-10-1 9 28 623 3574 23 622 1.255
rand0-20-10-10-20-2 9 6964 1540 48 538 1.265
rand-4-50-50-20-2 10 24 654 6162 98 610 12.379
rand-4-50-50-15-1 11 18 454 4622 119 367 8.76
rand-20-50-50-20-2 10 53 844 7047 168 793 28.348
rand0-4-10-10-20-1 11 5404 1360 770 565 19.168
rand-20-50-50-15-1 12 42 864 5357 2 716 507 358.345
rand-4-50-50-20-1 15 24 654 6162 39 265 111 7009.909

In Table 2, we depict the results of a larger evaluation on the datasets provided in [54]5,
which are also used in [16, 43]. We kept their file naming, which is the format rand-σ- m

2 - m
2 -n-i,

where i is an iteration counter to have multiple files with the same characteristics (m, n,
and σ). The prefix rand can be followed by a zero. We observe that larger distances correlate
with the number of choices, affecting the overall running time. Even for large inputs with
short distances like the dataset rand-4-150-150-5-1, the running time is short.

5 https://github.com/jeanpttorres/dssp

1 2 3 4 5 6 7 8 9 10 11 12 13

S1 = n s l l e e p e p l e s s

S2 = s l s e l e e p p l e s s

S3 = l s l e p e p n l e s n s

S4 = e s l e p e p l n e s l s

S5 = n s l e e p p l l e e s s

1 2 3 4 5 6 7 8 9 10 11 12 13

S1 = n s l l e e p e p l e s s

S2 = s l s e l e e p p l e s s

S3 = l s l e p e p n l e s n s

S4 = e s l e p e p l n e s l s

S5 = n s l e e p p l l e e s s

T = s l e e p l e s s

Figure 3 Example for lcs (Sect. 5) with n = 13. The input is shown on the left figure. In the
right figure, we highlighted the subsequences matching T = sleepless by surrounding the respective
characters with boxes in each input string. Here, T = sleepless is the lcs of all input strings.

5 Longest Common Subsequence (LCS)

The lcs problem asks for the longest string T such that T is a subsequence of Sx for every
x ∈ [1..m]. See Fig. 3 for an example.

Existence. A solution exists if all strings share at least one common character in the
alphabet.

Related Work. Maier [45] showed that lcs is NP-hard for σ ≥ 2, and the same holds for
scs with σ ≥ 5. Later, Blin et al. [5] gave a proof that lcs stays NP-hard even if the input
strings are well-compressible with the run-length encoding. For exact algorithms, we can
extend the classic dynamic programming (DP) algorithm of Wagner and Fischer [56] to m

strings, which then takes O(nm) time. Irving and Fraser [36] gave two algorithms running in
O(mn(n− ℓ)m−1) or O(mℓ(n− ℓ)m−1 + mσn) time, where ℓ is the length of the output. This
result implies that lcs is FPT in m and n − ℓ. Bulteau et al. [10] improved the result of [36]
with an algorithm running in O((n − ℓ + 1)n−ℓ+1mn) time, which is an FPT in the number
of deletions n − ℓ. Finally, there is a genetic algorithm [34] and an ant colony optimization
algorithm [50].

5.1 MAX-SAT encoding

Our idea is to select a subsequence Tx for each input string Sx and maximize the length of
Tx under the constraint that all Tx’s have to be equal. The subsequence Tx of Sx is given by
a sequence of indices i1 < . . . < i|Tx| such that Sx[i1] · · · Sx[i|Tx|] = Tx. We can encode the
subsequences Tx by the selectable variables Cx,ℓ,i encoding whether Tx[ℓ] = Sx[i], for each
x ∈ [1..m], ℓ ∈ [1..n]. We make use of Cx,ℓ,i as follows. First, for each Tx[ℓ], we define the
range for the selectable variables Cx,ℓ,i.6

[O(nm), O(n)] ∀x ∈ [1..m], ℓ ∈ [1..n] :
∑

i∈[ℓ..n]

Cx,ℓ,i ≥ 0 (LCS1)

6 Logically, we would expect in (LCS1) a “≤ 1” instead of a “≥ 0”. However, the former suffices together
with the following constraints and is cheaper than “≤ 1”.

If we have selected Tx[ℓ] to be Sx[i], then Tx[ℓ−1] must be a character chosen in Sx[1..i−1]:

[O(n2m), O(n)] ∀x ∈ [1..m], ℓ ∈ [2..n], i ∈ [ℓ..n] :

Cx,ℓ,i =⇒
∑

j∈[1..i−1]

Cx,ℓ−1,j = 1 (LCS2)

Next, we define the helper variables Vx,ℓ encoding whether Tx has a length of at least ℓ,
for each x ∈ [1..m], ℓ ∈ [1..n]. If we have selected a character for Tx[ℓ] via Cx,ℓ,i, then we set
Vx,ℓ to true to specify that Tx has a length of at least ℓ.

[O(nm), O(n)] ∀x ∈ [1..m], ℓ ∈ [1..n] :
∨

i∈[1..n]

Cx,ℓ,i =⇒ Vx,ℓ (LCS3)

We now restrict all Tx’s to be of equal length, which we do in a Round-Robin encoding:

[O(nm), O(1)] ∀x ∈ [1..m], ℓ ∈ [1..n] : Vx,ℓ =⇒ V(x+1) mod n,ℓ (LCS4)

Here, mod n : {1, 2, . . .} → [1..n] is the modulo operation with n mod n = n and (n +
1) mod n = 1. To achieve that all Tx store the same characters, we use the following
constraint.

[O(n3m), O(1)] ∀x ∈ [1..m], ℓ ∈ [1..n], i, j ∈ [1..n] :
Cx,ℓ,i ∧ C(x+1) mod m,ℓ,j =⇒ Sx[i] = S(x+1) mod m[j] (LCS5)

Finally, we enforce that we need to select a position for Tx[ℓ] if Vx,ℓ is set:

[O(nm), O(n)] ∀x ∈ [1..m], ℓ ∈ [1..n] : Vx,ℓ =⇒
∨

i∈[ℓ..n]

Cx,ℓ,i (LCS6)

Alternatively to (LCS5) and (LCS6), we can state that the next subsequence must select
one of the text positions j for Tx+1[ℓ] with Sx+1[j] = Sx[i].

[O(n2m), O(n)] ∀x ∈ [1..m], ℓ ∈ [1..n], i ∈ [1..n] :

Cx,ℓ,i =⇒
∑

j:Sx[i]=S(x+1) mod n[j]

C(x+1) mod m,ℓ,j = 1 (LCS5’)

Finally, we formulate our optimization problem as

[O(1), O(n)] maximize
∑

ℓ∈[1..n]

V1,ℓ (LCS7)

Complexities. Our implementation in ASP is given in Listing 3. We have O(mn2) selectable
variables (Cx,ℓ,i), O(mn) helper variables (Vx,ℓ), and O(n2m) clauses (LCS5’). The largest
clause has O(n) variables. An improvement for short lcs solutions could be to encode
the existence problem for a fixed length λ in ASP such that we have O(mλ) selectable
variables for encoding Tx, and call this encoding while varying λ to find the largest value for
λ admitting a solution.

CPM 2023



17:10 Encoding Hard String Problems with Answer Set Programming

Table 3 Evaluation of the Longest Common Subsequence problem (lcs).

ASP brute-force

file length rules vars choices time [s] choices time [s]
s02m11n023i1 10 166 538 21 494 23 617 1.00 8 388 608 47.29
s02m10n023i2 10 151 627 19 540 34 146 1.02 8 388 608 43.35
s02m09n023i1 11 137 112 17 586 10 964 0.61 8 388 608 39.73
s03m08n023i1 8 138 002 15 632 4831 0.40 8 388 608 39.20
s04m09n023i1 6 162 617 17 586 3927 0.39 8 388 608 39.07
s03m11n023i2 8 188 366 21 494 20 672 1.18 8 388 608 38.99
s03m08n023i2 7 136 795 15 632 11 046 0.63 8 388 608 38.54
s03m07n023i2 9 119 551 13 678 5945 0.40 8 388 608 37.59
s04m11n023i1 6 197 886 21 494 5767 0.58 8 388 608 37.06
s03m08n023i0 8 136 968 15 632 6301 0.45 8 388 608 37.05
s03m08n022i0 8 120 880 14 256 5467 0.37 4 194 304 17.87
s03m08n022i1 7 120 416 14 256 3970 0.32 4 194 304 17.69
s02m11n022i1 11 146 880 19 602 11 779 0.53 4 194 304 17.61
s03m07n022i2 9 105 785 12 474 2763 0.24 4 194 304 17.34
s03m11n022i2 7 165 908 19 602 7974 0.63 4 194 304 17.31
s04m11n022i1 6 175 570 19 602 8045 0.58 4 194 304 17.02
s02m09n022i1 12 121 186 16 038 6522 0.27 4 194 304 16.85
s03m08n022i2 8 120 313 14 256 4442 0.34 4 194 304 16.80
s04m09n022i1 6 143 324 16 038 5791 0.45 4 194 304 16.72
s02m10n022i2 10 135 128 17 820 9640 0.47 4 194 304 15.94

Listing 3 ASP for lcs (Sect. 5).

mat(X,I) :- s(X,I,_).
0 {c(X,L,I) : mat(X,I), I >= L} :- mat(X,L). %(LCS1)
1 {c(X,L,J) : J < I, mat(X,J)} 1 :- c(X,L+1,I), mat(X,L), mat(X,L+1). %(

↪→ LCS2)
v(X,L) :- c(X,L,I), mat(X,I), mat(X,L). %(LCS3)
v(X+1,L) :- v(X,L), mat(X,L), mat(X+1,L). %(LCS4)
v(0,L) :- v(m-1,L).
:- c(X+1,L,J), c(X,L,I), s(X,I,D), not s(X+1,J,D). %(LCS5)
:- c(0,L,J), c(m-1,L,I), s(m-1,I,D), not s(0,J,D).
1 {c(X,L,I) : mat(X,I), I >= L} :- v(X,L). %(LCS6)
#maximize {1,L : v(0,L)}. %(LCS7)
#show c/3.

5.2 Evaluation
A DP approach would need O(nm) time (cf. [15, Chapter IV, Section 15.4] for a textbook
reference). Here, we stick to a trivial approach that tries all distinct subsequences of the
first string S1, and for each such subsequence we check whether it is a subsequence of all
other input strings. The number of these subsequences is at most 2n − 1. If we select these
subsequences with respect to their lengths, starting with the longest possible one, we can
terminate whenever the selected subsequence is found in all other strings. In the worst

ref9,9

Figure 4 Example for mcsp (Sect. 6) with n = 13. We can factorize S1 = F1F2F3 into three
factors, with F1 = G3, F2 = G2 and F3 = G1 such that S2 = G1G2G3. Hence, the solution for this
example is a partition of length three. On the right is a partial assignment of the variable ref based
on this partition, where ref induces a factor starting at position 10 in S1.

case, the time complexity of this approach grows exponentially in n, but only linearly in m,
independent of the alphabet size. We therefore restrict our evaluation in Table 3 to scaling n

while keeping the other parameters unchanged. Like in Sect. 4.2, the ASP implementation
outperforms the brute-force approach. However, a DP implementation might outperform the
ASP implementation by re-using memoized results.

6 Minimum Common String Partition (MCSP)

For the special case of m = 2 input strings S1 and S2, the mcsp problem, introduced by
Goldstein et al. [29] and Swenson et al. [52], asks, for a given z ∈ [1..n], a factorization
of S1 into Sx = F1 · · · Fz and a permutation π of [1..z] such that Fπ(1) · · · Fπ(z) = S2. The
optimization problem is to find the smallest z for which a solution exists. We give an example
in Fig. 4.

Existence. A sufficient condition for whether a solution for any z ∈ [1..n] exists is to check
that the Parikh vectors of S1 and S2 are the same, such that at least a permutation on [1..n]
exist to rearrange the characters of S1 to form S2.

Related Work. While introducing mcsp, Goldstein et al. [29] also showed that it is NP-
hard. Bulteau and Komusiewicz [11] showed that mcsp is FPT in z. For constant alphabets
(σ = O(1)), Cygan et al. [17] presented an exact algorithm running in 2O(n lg lg n/ lg n) time.
Recently, Chromý and Sinnl [14] studied a DP algorithm. It is known that mcsp can
be tackled by probabilistic tree searches [7], ILP formulations [6, 23], and an ant colony
optimization algorithm [22].

6.1 MAX-SAT encoding

We adapt the MAX-SAT encoding of Bannai et al. [1] for the shortest bidirectional macro
scheme problem [51]. To this end, we define the sets Mi := {j ∈ [1..n] | S1[i] = S2[j]} ⊂ [1..n]
specifying the positions in S2 that match with S1[i]. In what follows, we make use of the
following selectable Boolean variables:

pi or qi encode if S1[i] or S2[i] is the start of a factor, respectively, for i ∈ [1..n].
ref i→j encodes whether position i of S1 references position j of S2, and vice versa, for
i ∈ [1..n] and j ∈ Mi.

CPM 2023

We have O(n2) Boolean variables, which we use as follows. On the one hand, each position
in S1 has exactly one reference:

[O(n), O(n)] ∀i ∈ [1..n] :
∑

j∈Mi

ref i→j = 1 (MCSP1)

On the other hand, each position in S2 has exactly one reference:

[O(n), O(n)] ∀j ∈ [1..n] :
∑

i∈[1..n]

ref i→j = 1 (MCSP2)

In what follows, we add implications for the factor starting positions that are due to how
we set the references. First, a factor starts always at the first text position, so p1 and q1 are
always true. If S1[i] references S2[i] and i is a factor starting position of S1, so is j for S2.

[O(n2), O(1)] ∀i ∈ [1..n], j ∈ Mi : pi ∧ ref i→j =⇒ qj (MCSP3)

Next, if S1[i] references S2[i] and j is a factor starting position of S2, so is i for S1. We only
have to check that condition for q1 since all other constraints set pi and constraint (MCSP3)
then implies that qj has to be set.

[O(n), O(1)] ∀i ∈ [1..n] : q1 ∧ ref i→1 =⇒ pi (MCSP4)

Another condition is that if the previous text positions have mismatching characters, we
cannot extend the factor to the left.

[O(n2), O(1)] ∀i ∈ [1..n], j ∈ Mi with S1[i − 1] ̸= S2[j − 1] : ref i→j =⇒ pi (MCSP5)

Even if the previous characters match, when the reference of the previous text positions
is different, we need to make a factor starting position:

[O(n2), O(1)] ∀i ∈ [2..n], ∀j ∈ Mi such that j > 1 and S2[i − 1] = S2[j − 1] :
¬ref i−1→j−1 ∧ ref i→j =⇒ pi (MCSP6)

[O(1), O(n)] Finally, we minimize
∑

i∈[1..n]

pi (MCSP7)

Complexities. We have O(n2) selectable variables, and O(n2) clauses (MCSP3). The largest
clause has O(n) variables (MCSP2). Our implementation in ASP is given in Listing 4. Note
that we start counting at zero, so p(0). is equivalent to setting p1 to true. Instead of mat we
use the helper variables spos and tpos denoting the existence of S1[i] and S2[i], respectively.

Listing 4 ASP for mcsp (Sect. 6).

spos(I) :- s(0,I,_).
tpos(J) :- s(1,J,_).
p(0). q(0).
arc(I,J) :- s(0,I,C), s(1,J,C).
1 {ref(I,J) : arc(I,J)} 1 :- spos(I). %(MCSP1)
1 {ref(I,J) : arc(I,J)} 1 :- tpos(J). %(MCSP2)
q(J) :- p(I), ref(I,J). %(MCSP3)
p(I) :- q(1), ref(I,1). %(MCSP4)
p(I) :- ref(I,J), s(0,I-1,C), s(1,J-1,D), C != D. %(MCSP5)
p(I) :- not ref(I-1,J-1), ref(I,J). %(MCSP6)
#minimize {1,X : p(X)}. %(MCSP7)
#show ref/2. #show p/1. #show q/1.

Table 4 Evaluation of the Minimum Common String Partition problem (mcsp). Note that
the time for the ASP solution is in milliseconds. The column z denotes the number of factors of the
returned partition.

ASP brute-force

file z rules vars choices time [ms] choices time [s]
2s03n009i2 4 443 124 25 1.0 986 409 6.24
2s02n009i0 4 586 165 61 2.0 986 409 6.31
2s02n009i1 4 586 165 59 2.0 986 409 6.43
2s03n009i0 6 426 124 52 1.0 986 409 6.44
2s03n009i1 2 367 116 30 1.0 986 409 6.49
2s02n009i2 6 521 149 39 1.0 986 409 6.95
2s03n010i1 4 604 162 67 2.0 9 864 100 68.81
2s02n010i0 4 510 213 108 2.0 9 864 100 70.92
2s03n010i0 6 484 147 37 1.0 9 864 100 71.04
2s03n010i2 4 584 164 47 2.0 9 864 100 71.38
2s02n010i2 4 637 189 77 2.0 9 864 100 73.78
2s02n010i1 3 639 187 103 2.0 9 864 100 74.28

Table 5 Evaluation of the Minimum Common String Partition problem (mcsp) on prefixes of
the SARS-CoV-2 dataset.

length z rules vars choices time [s]
10 4 447 146 34 0.001
20 12 1273 445 269 0.003
30 14 2282 911 1951 0.017
40 16 3720 1685 4683 0.047
50 21 5468 2442 2 050 092 18.609
60 24 7451 3422 6 866 999 80.256

6.2 Evaluation

Without leveraging the actual contents of the characters like in our SAT formulation, a
naive way is to factorize both strings S1 and S2 with factors of the same lengths, and check
whether there exists a permutation such that we can match factors of S1 with factors of S2.
To this end, we iterate over the size z of the partition from 1 to n. For each z ∈ [1..n], we
partition S1 into z factors S1 = F1 · · · Fz. There are

(
n
z

)
such ways to partition S1. For each

permutation πz on [1..z], we define the factorization G1 · · · Gz = S2 with |Gx| = |Fπ(x)| for
all x ∈ [1..z]. If Gx = Fπ(x) for all x ∈ [1..z], then we have found a solution, and terminate.
The number of configurations is

∑n
z=1

(
n
z

)
z! , and each check takes O(n) time. Like the

brute-force approach for lcs (Sect. 5.2), this approach has an exponential dependency on
the text length n. In Table 4, we observe that specifying the choices for the references for
each position individually (as we do in our ASP encoding) reduces the number of choices
significantly when compared to the choices the brute-force algorithm processes.

Since our ASP encoding for mcsp seems quite efficient, we subsequently performed a
benchmark on real data. In detail, we conducted an experiment by scaling the prefix length
of a given input sequence, and report results in Table 5. For that, we used the SARS-CoV-2

1 2 3 4 5

S1 = e e p l e

S2 = l e s s n

S3 = p l e s s

S4 = s l e e p

S2 = l e s s n

S3 = p l e s s

T = s l e e p l e s s n e s s

Figure 5 Example for scs (Sect. 7) with n = 5. The input is shown on the left figure. By the right
figure, the scs is T = sleeplessness, where we shifted the input strings to match their occurrences
in T .

reference in FASTA format introduced in the analysis of Farkas et al. [21] 7, after removing
the header line and the newline characters. For each extracted prefix of this FASTA file,
we created an instance for mcsp, where the second string is a random permutation of the
original prefix. We can observe in Table 5 that the output size z exponentially correlates
with the number of choices and the running time.

7 Shortest Common Superstring (SCS)

The scs problem asks for the shortest string T such that Sx is a substring of T , for all
x ∈ [1..m]. Figure 5 shows an example.

Existence. A trivial common superstring is the concatenation S1 · · · Sm. Permuting the
strings and removing overlapping parts lead to the solution [25].

Related Work. Gallant et al. [25] showed that scs is NP-hard for n ≥ 3 with respect to the
number of strings m and unbounded alphabet size, but can be solved in linear time if n ≤ 2.
For binary alphabet σ = 2, they showed that the problem is still NP-hard for n = Ω(log(nm)).
It is known that scs can be solved with neural networks [44] and genetic algorithms [30].
Most research on scs is devoted to the analysis and improvement of the approximation
algorithm presented by Tarhio and Ukkonen [53, Theorem 2.3]. This algorithm builds the
so-called overlap graph of S. The authors observed that a Hamiltonian path on the overlap
graph [49] maximizing the weights of the selected edges solves scs.

7.1 Reduction to Hamiltonian Path
We follow the idea of Tarhio and Ukkonen [53] by reducing scs to the search of the Hamiltonian
path maximizing the weights of the selected edges. The ASP encoding of finding a Hamiltonian
cycle in an unweighted graph has already been studied in [42, 41]. We build on one of their
approaches and extend it by maximizing the weights while omitting the weight of one
edge to turn the cycle into a Hamiltonian path8. An overlap graph (S, A, w) is a weighted

7 https://github.com/cfarkas/SARS-CoV-2-freebayes
8 We make a distinction between Hamiltonian path and Hamiltonian cycle in the sense that the cycle

visits exactly one node twice.

https://github.com/cfarkas/SARS-CoV-2-freebayes

directed graph, having the input strings S as nodes and the arcs A := {(Sx, Sy) : x ̸= y}.
The weights are defined by a weight function w : A → [0..n] with w(Sx, Sy) := max{|U | :
U is suffix of Sx and prefix of Sy}. Hence, w(Sx, Sy) is the number of overlapping characters,
which we can omit if we want to build the superstring of Sx and Sy that starts with Sx.
With respect to the overlap graph, a path is a sequence of strings, and a Hamiltonian path in
the overlap graph is a path that visits each node exactly once, i.e., a permutation π of the
list [S1, . . . , Sm]. Our goal is to find a permutation that maximizes

∑m−1
x=1 w(Sπ(x), Sπ(x+1)),

i.e., to find the Hamiltonian path whose arcs have maximal weights in sum.

7.2 MAX-SAT encoding
We define the following O(m2) Boolean variables:

cyclex,y encoding whether we have the arc (Sx, Sy) in our Hamiltonian cycle, for x, y ∈
[1..m];
reachx,y encoding whether we can reach Sy from Sx by following the transitive closure of
cycle, for x, y ∈ [1..m];
startx encoding whether our superstring starts with Sx, for x ∈ [1..m].

First, we select arcs from the overlap graph for cyclex,y. To this end, for each string Sx,
we select exactly one out-going arc and one in-coming arc:

[O(m), O(m)] ∀x ∈ [1..m] :
m∑

y=1
cyclex,y = 1 and ∀y ∈ [1..m] :

m∑
x=1

cyclex,y = 1 (SCS1)

The transitive closure of cycle can be encoded as follows. First we initialize reach by the
direct connections due to cycle.

[O(m2), O(1)] ∀x, y ∈ [1..m], x ̸= y : cyclex,y =⇒ reachx,y (SCS2)

Next, if we can reach y from x, and there is an arc (y, z), then we can reach z from x:

[O(m3), O(1)] ∀x, y, z ∈ [1..m], x ̸= y ̸= z : reachx,y ∧ cycley,z =⇒ reachx,z (SCS3)

To make the path selected by cyclex,y an Hamiltonian path, we want that all strings are
connected via reach:

[O(m2), O(1)] ∀x, y ∈ [1..m], x ̸= y : reachx,y = 1 (SCS4)

For the Hamiltonian path it is left to select a designated start string9.

[O(1), O(m)]
m∑

y=1
starty = 1 (SCS5)

Finally, our objective is to maximize the weights on the path starting from startx of
length m:

[O(1), O(m2)] maximize
∑

x,y∈[1..m]: cyclex,y∧¬starty

w(x, y) (SCS6)

9 It actually suffices to check in (SCS4) that all strings can be reached from this start string, but doing so
had a negative impact on the overall running time in the experiments.

Table 6 Evaluation of the Shortest Common Superstring problem (scs). |T | is the length of
the SCS.

ASP brute-force

file |T | rules vars choices time [s] choices time [s]
s02m10n008i0 42 2090 1416 198 756 3.58 10 240 0.02
s02m10n008i1 33 2206 1465 1 854 941 40.73 10 240 0.02
s02m10n008i2 39 2200 1464 1 401 686 29.49 10 240 0.02
s02m11n008i0 49 2639 1825 2 150 681 44.96 22 528 0.03
s02m11n008i1 35 2699 1861 6 652 411 154.48 22 528 0.03
s02m11n008i2 50 2611 1817 6 980 725 136.00 22 528 0.02

Complexities. We have O(m2) selectable variables and O(m3) clauses (SCS3). The largest
clause has O(m2) variables (SCS6). Our implementation in ASP is given in Listing 5. We
expect an input of the form w(X,Y,C) encoding the weight w(X, Y) = C. The helper variables
node and gain define the nodes of the overlap graph and the value of the optimization
argument in (SCS6), respectively.

Listing 5 ASP for scs (Sect. 7).

node(X) :- w(X,_,_).
1 {cycle(X,Y) : w(X,Y,_)} 1 :- node(X). %(SCS1)
1 {cycle(X,Y) : w(X,Y,_)} 1 :- node(Y).
reach(X,Y) :- cycle(X,Y). %(SCS2)
reach(X,Z) :- reach(X,Y), cycle(Y,Z). %(SCS3)
:- not reach(X,Y), node(X), node(Y). %(SCS4)
1 {start(X) : node(X)} 1. %(SCS5)
gain(D) :- D = #sum {C,X : cycle(X,Y), w(X,Y,C), not start(Y)}. %(SCS6)
#maximize {D : gain(D)}.
#show cycle/2. #show start/1.

7.3 Evaluation

The overlap graph can be computed in O(nm + m2) time [33]. Given the overlap graph, the
easiest approach is to enumerate all m! permutations, and compute the sum of the selected
weights in Θ(m) time. The time bound can be improved by using a DP approach taking
O(m22m) time10. In the experiments of Table 6, we use this DP approach as our brute-force
solution. We observe that it outperforms our ASP implementation on all instances. That is
due to the fact that (a) our ASP encoding does not make use of more information than the
DP approach, and that (b) the number of choices in our encoding for the Hamiltonian path
is prohibitively large. As a matter of fact, efficient SAT and ASP encodings for Hamiltonian
cycles are actively studied, cf. [58] for SAT and [4] for ASP.

10 https://leetcode.com/problems/find-the-shortest-superstring/solutions/194891/
official-solution/

https://leetcode.com/problems/find-the-shortest-superstring/solutions/194891/official-solution/
https://leetcode.com/problems/find-the-shortest-superstring/solutions/194891/official-solution/

Table 7 Encoding complexities of the studied problems. Columns prob., #sel. vars, #h.vars,
#clauses and max. cl. denote the problem name, the number of defined selectable variables, the
number of helper variables, the number of clauses, and the maximum size a clause can have.

prob. #sel.vars #h.vars #clauses max cl.
csp O(nσ) O(nm) O(nmσ) O(mn)
css O(λσ+(n−λ)m) O(λm) O(nmσλ) O(λm)
lcs O(n2m) O(mn) O(n2m) O(n)
mcsp O(n2) O(1) O(n2) O(n)
scs O(m2) O(1) O(m3) O(m2)

8 Conclusion

We provided encodings in ASP for five prominent examples of NP-hard problems in the field
of stringology. We summarized the complexities of the encodings in Table 7. We observed
that, on the one hand, by leveraging characteristics of the input data such as for mcsp, our
solution is far superior than simple brute-force approaches that omit those characteristics. On
the other hand, for scs, we observed that if the problem can be easily reduced to instances
of problems like finding a Hamiltonian path, DP approaches are already efficient enough
to find the answer faster than an ASP solver. It therefore depends on the nature of the
problem we study for whether an application of an ASP solver makes sense. Nevertheless,
the programming in ASP is highly expressive as can be seen by the short program codes
in Listings 1–5, and therefore can be understood as a tool for rapid prototyping. Other
advantages of ASP solvers like clingo are that they can work in parallel, report approximate
solutions when reaching a given timeout, and enumerate all solutions, provided that the
specified constraints do not exclude one of them. An evaluation of those features is left as
future work since it would go beyond the scope of this paper.

A Alternative CSS Encoding

For small values of λ, the offsets can be quite large. Here, we present an alternative encoding
without the offsets. The resulting encoding has fewer variables, but has more variables that
are subject to the optimization argument. In what follows, we can encode T [1..λ] by the
Boolean variables Ti,c specifying with Ti,c = 1 that T [i] = c:

[O(λ), O(σ)] ∀i ∈ [1..λ] :
∑
c∈Σ

Ti,c = 1 (CSS1’)

We now let the costs encode the offsets by the variables Ci,x,o being set if Sx[o + i] ̸= T [i].

[O(λnmσ), O(1)] ∀i ∈ [1..λ], c ∈ Σi, x ∈ [1..m], o ∈ [1..n − λ] :
Ti,c ∧ Sx[i + o] ̸= c =⇒ Ci,x,o (CSS2’)

Table 8 Used Entities.

entity meaning

Σ alphabet
σ alphabet size, σ = |Σ|
S set of input strings {S1, . . . , Sm}
m size of S, i.e., m = |S|
n length of an input string
Sx input string
T string to output
ℓ length for a subsequence
δ distance of the output to all Sx

i, j indices for text positions in an input string
x, y indices for an input string
c character in Σ

The objective function becomes

[O(1), O(mn2)] minimize max
x∈[1..m]

min
o∈[1..n−λ]

∑
i∈[1..n]

Ci,x,o (CSS3’)

